
Anja Aue | 07.02.17

Bedingung
 nein

Anweisung

ja

Anweisung

V(isual) B(asic for) A(pplication)
Bedingte Anweisungen



Visual Basis for Application Seite 2

Bedingte Anweisung

    Dim temperatur As Double
    
    temperatur = 0.6
    
    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
      
    ElseIf (temperatur < 0#) Then
       Debug.Print "Minus-Temperatur"
    
    Else
        Debug.Print "Null-Punkt"
        
    End If 

Wenn-Dann-Anweisungen

 Wenn die Temperatur einen bestimmten Wert erreicht hat, 
dann …

 Wenn das Datenfeld A den Wert x gespeichert hat, dann ...
 Wenn der Kontostand dem Dispo übersteigt, dann ...
 Wenn die Strecke A doppelt so lang ist wie Strecke B, dann ...
 Wenn die Warenmenge eine Mindestmenge unterschreitet, 

dann ...

if-Anweisung

Dim temperatur As Double
    
    temperatur = 0.6
    
    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
    End If

 Wenn (if) die Bedingung zutrifft, dann (then) führe die 
Anweisungen aus ...

 Wenn die Bedingung war ist, werden die darunter stehenden 
Anweisungen ausgeführt.

Arbeitsweise

If (temperatur > 0#) Then

Debug.Print "Plus-Temperatur" 

End If

Bedingung = true

Aufbau

If [Bedingung] Then

[Anweisung]

End If

[Anweisung]

Kopf der if-Anweisung

    
 If (temperatur > 0#) Then

 Beginn mit dem Schlüsselwort If.
 Dem Schlüsselwort folgt die Bedingung. Die Bedingung kann, 

muss aber nicht geklammert werden.
 Der Kopf der Anweisung endet mit dem Schlüsselwort Then.

Rumpf der if-Anweisung

    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"

 Wenn die Bedingung wahr ist, werden die Anweisungen die, 
dem Schlüsselwort Then folgen, ausgeführt.

 Die Anweisungen sollten entsprechend der if-Anweisung 
eingerückt werden.

Ende der if-Anweisung

    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
    End if

 Die if-Anweisung endet mit den Schlüsselwörtern End If.

Bedingungen 

 Ausdrücke, die einen boolschen Wert zurückgeben.
 Vergleich von Werten. Beantwortung von Fragen wie zum 

Beispiel „Ist der Messwert größer als 0“. 



Visual Basis for Application Seite 11

Beispiel

    temperatur = 0.6
    
    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"

 Mit Hilfe von Vergleichsoperatoren werden zwei Werte 
verglichen. In diesem Beispiel wird der Wert in der Variablen 
temperatur mit Null verglichen.

 Vergleichsoperatoren können aus ein oder zwei Zeichen 
bestehen. Zwischen den Zeichen darf kein Leerzeichen stehen.

Vergleichsoperatoren

5Ist kleiner < 2 =

5Ist kleiner gleich <= 2 =
var

false

5Ist größer > 2 =
var
true

var
false

5Ist größer gleich >= 2 =
var
true

Vergleichsoperatoren „Ist gleich“ und „Ist ungleich“

5Ist gleich = 2 =

5Ist nicht gleich <> 2 =
var
true

var
false



Visual Basis for Application Seite 14

Beispiel: Ist gleich

       If (rs.Fields("Auslaufartikel").Value = True) Then
            
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist ein Auslaufartikel."
        End If

 Das Feld in der Tabelle ist vom Datentyp Boolean. Ist die 
Aussage „Ist das Produkt ein Auslaufartikel?“ wahr?

 Ist das Kontrollkästchen aktiv?
 Entspricht die Variable exakt dem angegeben Wert? 

Gleitkommazahlen sollten nicht auf Gleichheit überprüft 
werden.

Beispiel: Ist ungleich

       If (rs.Fields("Auslaufartikel").Value <> True) Then
            
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist ein Auslaufartikel."
        End If

 Das Feld in der Tabelle ist vom Datentyp Boolean. Ist die 
Aussage „Ist das Produkt ein Auslaufartikel?“ falsch?

 Ist das Kontrollkästchen deaktiviert?
 Entspricht die Variable nicht dem angegebenen Wert?



Visual Basis for Application Seite 16

Leere Zeichenfolge

    Dim wort As String
    
    If (wort = "") Then
        wort = "abcd"

    End If

 Der String hat einen definierten Wert.
 Wenn die Bedingung wahr ist, enthält die Variable wort eine 

leere Zeichenfolge.

Undefiniertes Datenfeld

    rs.MoveFirst

    Do While Not (rs.EOF)
        
        If IsNull(rs.Fields("Versanddatum").Value) Then
            
            bestellnummer = rs.Fields("Bestell-Nr").Value()
            Debug.Print "Keine Versendung"
        End If
        
        rs.MoveNext
    Loop

Erläuterung

 Der Funktion IsNull() wird das zu untersuchende Element 
übergeben.

 Wenn das Datenfeld einen undefinierten Inhalt enthält, ist die 
Bedingung wahr.

Is Nothing

    
     If dbs Is Nothing Then
        Debug.Print "Kein Verweis vorhanden"
        Exit Sub
    End If
  

 Ist die Objekt-Variable leer?
 Wenn kein gültiger Verweis vorhanden ist, wird die Prozedur 

mit Exit Sub abgebrochen.

… andernfalls

    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
    Else
        Debug.Print "Null-Punkt"
    End If

 Andernfalls (Else)  führe … aus.
 Beschreibung des Standardfalls.
 Else-Anweisungen beziehen sich immer auf eine if-Anweisung.
 Die else-Anweisung und die dazugehörige Anweisung haben 

die gleiche Einrücktiefe.
 Die else-Anweisung ist optional.

Grafische Darstellung

temperatur > 0

Anweisung

Anweisungnein

ja

Fallunterscheidung

    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
      
    ElseIf (temperatur < 0#) Then
       Debug.Print "Minus-Temperatur"
    
    Else
        Debug.Print "Null-Punkt"
        
    End If

Grafische Darstellung

temperatur > 0

Anweisung

Temperatur < 0

Anweisung

nein

ja ja

nein



Visual Basis for Application Seite 24

Erläuterung

 Im ersten Schritt wird die Bedingung der if-Anweisung 
überprüft.

 Falls diese nicht zutrifft, wird die erste else if-Anweisung 
überprüft. Wenn die Bedingung zutrifft, wird der dazugehörige 
Anweisungsblock ausgeführt. Alle nachfolgenden Bedingungen 
sowie die else-Anweisung werden ignoriert.

 Falls diese nicht zutrifft, wird die nächste else if-Anweisung 
überprüft. 

 Falls keine der Bedingungen zutrifft, wird, falls vorhanden, die 
else-Anweisung ausgeführt.

Bedingte Anweisungen verschachteln

  If IsNull(rs.Fields("Versanddatum").Value) Then
            
        maxVersand = DateAdd("m", 3, rs.Fields("Bestelldatum").Value)
           
         If (Date > maxVersand) Then
                bestellnummer = rs.Fields("Bestell-Nr").Value()
                Debug.Print "Keine Versendung"
          End If

  End If

Logische Operatoren

 Verknüpfungen von Ausdrücken, die einen boolschen Wert 
zurückliefern.

 Und-Verknüpfung. Beide Bedingungen müssen erfüllt sein.
 Oder-Verknüpfung. Eine der beiden Bedingungen müssen 

erfüllt sein.
 Nicht. Der Wert der Bedingung wird negiert.

Und-Verknüpfung

Beispiel „und“

    If (temperatur >= 36.9) And (temperatur <= 37.4) Then
        Debug.Print "Erhöhte Temperatur"
        
    ElseIf (temperatur >= 37.5) And (temperatur <= 39.4) Then
        Debug.Print "Fieber"
    ElseIf (temperatur >= 39.5) And (temperatur <= 40.5) Then
        Debug.Print "Hohes Fieber"
    ElseIf (temperatur >= 40.5) Then
        Debug.Print "Sehr hohes Fieber"
    
    Else
        Debug.Print "Normaltemperatur"
    End If

Hinweise

 If (temperatur >= 36.9) And (temperatur <= 37.4) Then
        Debug.Print "Erhöhte Temperatur"
        
    ElseIf (temperatur >= 37.5) And (temperatur <= 39.4) Then
        Debug.Print "Fieber"

 End If

 Wenn die linke Bedingung wahr ist, wird die rechte Bedingung 
überprüft.

 Werte zwischen 37.4 und 37.5 werden nicht abgefangen.

Oder-Verknüpfung

Beispiel „oder“

    If (monat = 12) Or (monat = 1) Or (monat = 2) Then
        Debug.Print "Winter"
        
    ElseIf (monat = 3) Or (monat = 4) Or (monat = 5) Then
        Debug.Print "Frühjahr"
    ElseIf (monat = 6) Or (monat = 7) Or (monat = 8) Then
        Debug.Print "Sommer"
    ElseIf (monat = 9) Or (monat = 10) Or (monat = 11) Then
        Debug.Print "Herbst"

    Else
        Debug.Print "Keine korrekte Monatsangabe"
    End If

Hinweise

  If (monat = 12) Or (monat = 1) Or (monat = 2) Then
        Debug.Print "Winter"
        
    ElseIf (monat = 3) Or (monat = 4) Or (monat = 5) Then
        Debug.Print "Frühjahr"

 Wenn die linke Bedingung falsch ist, wird die rechte Bedingung 
überprüft.

Nicht-Verknüpfung

not

not

false

true

=

=

true

false

Not Is Nothing

    
If Not (rs Is Nothing) Then

        rs.MoveLast
End If

  

 Der Operator Not negiert die Bedingung.
 Durch den Operator wird True zu False und umgekehrt.

Not

Vergleiche von Strings

    
       If (rs.Fields("Kategorie").value() Like "Getränke") Then
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()

        End If
  

 Mit Hilfe des Schlüsselwortes LIKE können Strings mit einem 
Textmuster verglichen werden.

 Textmuster beginnen und enden mit den Anführungsstrichen.

Platzhalter „Sternchen“

  
  If (rs.Fields("Produktcode") Like "NWTCO-*") Then

            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()

        End If
  

 Das Sternchen steht für eine beliebige Anzahl von Zeichen 
oder keines.

 Das Sternchen kann an jeder beliebigen Position im Textmuster 
stehen.

Platzhalter „Fragezeichen“

  
  If (rs.Fields("Produktcode") Like "NWT?-*") Then

            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            
        End If
  

 Das Fragezeichen steht für exakt ein alphanumerisches oder 
numerisches Zeichen.

 Das Fragezeichen kann an jeder beliebigen Position im 
Textmuster stehen.

Platzhalter „Liste von Zeichen“

  
 If (strNachname Like "[AB]*") Then

             Debug.Print strNachname 
            
        End If
  

 Mit Hilfe der eckigen Klammern wird eine Liste von Zeichen 
zusammengefasst.

 Die Zeichen dürfen an dieser Position vorkommen.
 In diesem Beispiel darf der Nachname mit dem Buchstaben A 

oder B beginnen.

Negieren des Platzhalter „Liste von Zeichen“

  
 If (strNachname Like "[!AB]*") Then

             Debug.Print strNachname 
            
        End If
  

 Mit Hilfe der eckigen Klammern wird eine Liste von Zeichen 
zusammengefasst.

 Das Aufrufezeichen negiert die Liste. Die Zeichen dürfen an 
dieser Position nicht vorkommen.

 In diesem Beispiel darf der Nachname nicht mit dem 
Buchstaben A oder B beginnen.

„Wähle aus“

               Select Case rs.Fields("Status-Nr").Value
                    Case 0
                        Debug.Print "neu angelegt."
                    Case 1
                        Debug.Print "in Rechnung gestellt."
                    Case 2
                        Debug.Print "versandt."
                    Case 3
                        Debug.Print "geschlossen."
                    Case Else
                        Debug.Print "ohne Statusangabe."
                    
                End Select

Aufbau

Select [variable] Case

End Select

Case Bedingung :

Anweisung 

Case Else Bedingung :

Anweisung 

Kopf der Select Case - Anweisung

 Den Schlüsselwörtern Select Case folgt der Name der zu 
untersuchenden Variable.

 Variablen vom Typ „String“, „Zahl“ oder „Datums-/Zeitwerte“ 
können untersucht werden.

Select intKategorie Case

Select [variable] Case

Beispiel: Überprüfung von Datumsangaben

    Dim aktuellDatum As Date
    aktuellDatum = #12/24/2017#
    
    Select Case aktuellDatum
        Case #12/24/2017#
            Debug.Print "Heiligabend"
        Case #12/25/2017#
            Debug.Print "1. Weihnachtstag"
        Case #12/26/2017#
            Debug.Print "2. Weihnachtstag"
        Case #12/31/2017#
            Debug.Print "Silvester"
        Case #1/1/2018#
            Debug.Print "Neujahr"
    End Select



Visual Basis for Application Seite 46

Fälle

 Den Schlüsselwörtern Case folgt ein Wert.
 Jeder Fall entspricht der Bedingung ([variable] = [wert]).
 Die einzelnen Fälle werden von oben nach unten durchlaufen.
 Die Werte der verschiedenen Fälle sind vom gleichen Datentyp 

der zu überprüfenden Variable oder können in diesen 
konvertiert werden.

Case #12/24/2017# 

Case [wert] 

Standardfall

 Die Schlüsselworte Case Else beschreiben den Standardfall.
 Der Standardfall wird ausgeführt, wenn keine anderen Fälle 

zutreffen.
 Der Fall Case Else ist optional.

Beispiel: Nutzung einer Auflistungen

    Dim land As String
    land = "Polen"
    
    Select Case land
        Case "Polen", "Dänemark", "Niederlande"
            Debug.Print "Europa"
        
        Case "Indien", "China"
            Debug.Print "Asien"

    End Select

Erläuterung

 Den Schlüsselwörtern Case folgt eine Auflistung von Werten.
 Die Werte in der Auflistung werden durch ein Kommata 

getrennt.
 Jeder Fall entspricht der Bedingung 

([variable] = [wert]) OR ([variable] = [wert]).

Case "Indien" , "China"

Case [wert] , [wert]

Beispiel: Nutzung von Wertebereichen

    Select Case monat
        Case 3 To 5
            Debug.Print "Frühling"
        
        Case 6 To 8
            Debug.Print "Sommer"
            
        Case 9 To 11
            Debug.Print "Herbst"

    End Select

Erläuterung

 Den Schlüsselwörtern Case folgt ein Wertebereich.
 Der Wertebereich wird durch einen unteren und oberen Wert 

begrenzt.
 Jeder Fall entspricht der Bedingung 

([variable] >= [wert]) AND ([variable] <= [wert]).

Case 4 To 5

Case [min] To [max]

Beispiel: Nutzung von Vergleichsoperatoren

   Select Case temperatur
        Case 36.9 To 37.4
            Debug.Print "Erhöhte Temperatur"
        
        Case Is > 37.4 And temperatur <= 39.4
            Debug.Print "Fieber"
        
        Case Is > 39.4 And temperatur <= 40.5
            Debug.Print "Hohes Fieber"
        
        Case Is > 40.5
            Debug.Print "Sehr hohes Fieber"
    End Select



Visual Basis for Application Seite 53

Erläuterung

 Den Schlüsselwörtern Case folgt eine Bedingung.
 Das Schlüsselwort Is ist ein Platzhalter für die zu überprüfende 

Variable.

Case Is > 40.5

Case Is [operator] [wert]

Verknüpfung von Bedingungen

 Die erste Bedingung beginnt mit Case Is. Dem Schlüsselwort Is  
folgt der Vergleichsoperator und der Vergleichswert.

 Bedingungen können mit AND oder OR verknüpft werden.
 In der zweiten Bedingung muss der Name der zu 

überprüfenden Variablen genutzt werden.

   Select Case temperatur
        Case Is > 37.4 And temperatur <= 39.4
            Debug.Print "Fieber"
    End Select

Hinweis

 Der oben dargestellte Fall ist immer wahr.
 Die Fallunterscheidung entspricht einem Zahlenraum. Ein 

Zahlenraum wird folgendermaßen dargestellt: 36.9 To 37.4.

   Select Case temperatur

        Case Is >= 36.9 And temperatur <= 37.4
            Debug.Print "Erhöhte Temperatur"

    End Select

Beispiel: Nutzung von Vergleichsoperatoren

   Select Case temperatur
        Case Is >= 36.9 And (temperatur <= 37.4)
            Debug.Print "Erhöhte Temperatur"
        
        Case Is > 37.4 And temperatur <= 39.4
            Debug.Print "Fieber"
        
        Case Is > 39.4 And temperatur <= 40.5
            Debug.Print "Hohes Fieber"
        
        Case Is > 40.5
            Debug.Print "Sehr hohes Fieber"
    End Select

Attribute VB_Name = "anweisung_if"
Option Compare Database
Option Explicit

Sub temperaturBerechnung()
    Dim temperatur As Double
    
    temperatur = 0.6
    
    If (temperatur > 0#) Then
        Debug.Print "Plus-Temperatur"
      
    ElseIf (temperatur < 0#) Then
       Debug.Print "Minus-Temperatur"
    
    Else
        Debug.Print "Null-Punkt"
        
    End If
End Sub

Sub vergleichBoolean()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim strProdukt As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    Do While Not (rs.EOF)
        
        If (rs.Fields("Auslaufartikel").Value = True) Then
            
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist ein Auslaufartikel."
        End If
        
        If (rs.Fields("Auslaufartikel").Value <> True) Then
            
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist kein Auslaufartikel."
        End If
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub StringsWithLike()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim strProdukt As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    Do While Not (rs.EOF)
            
        If (rs.Fields("Kategorie").Value() Like "Getränke") Then
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist ein Getränk."
            
        End If
        
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub StringsWithLikeAndPlaceholder()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim strProdukt As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    Do While Not (rs.EOF)
        
        If (rs.Fields("Produktcode").Value() Like "NWTCO-*") Then
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " beginnt mit NWTCO."
            
        End If
        
        If (rs.Fields("Produktcode").Value() Like "NWT?-*") Then
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " beginnt mit NWT plus einen Buchstaben."
            
        End If
        
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub StringsWithLikeAndSetOfCharacter()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim strNachname As String
    
    tblName = "Kunden"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    Do While Not (rs.EOF)
            
        strNachname = rs.Fields("Nachname").Value
        
        If (strNachname Like "[AB]*") Then
             Debug.Print strNachname & " arbeitet in der Firma " & rs.Fields("Firma").Value
            
        End If
        
        If (strNachname Like "[!AB]*") Then
             Debug.Print strNachname & " arbeitet in der Firma " & rs.Fields("Firma").Value
            
        End If
        
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub
Sub suchenFinden()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim strProdukt As String
    
    tblName = "Artikel"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenDynaset)
    rs.FindFirst "Kategorie LIKE 'Getränke'"
    
    If rs.NoMatch Then
        Debug.Print "Keine Getränke vorhanden."
        
    Else
        Do While Not (rs.NoMatch)
            
            strProdukt = rs.Fields("Produktcode").Value()
            strProdukt = strProdukt & ": " & rs.Fields("Artikelname").Value()
            Debug.Print strProdukt & " ist ein Getränk."
            
            rs.FindNext ("[Kategorie] LIKE 'Getränke'")
        Loop
    
    End If
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
End Sub

Sub vergleichStringIsLeer()
    Dim wort As String
    
    If (wort = "") Then
        wort = "abcd"
        Debug.Print wort
        
    End If
    
End Sub

Sub vergleichDatenfeldIsNull()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim bestellnummer As String
    
    tblName = "Bestellungen"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    Do While Not (rs.EOF)
        
        If IsNull(rs.Fields("Versanddatum").Value) Then
            
            bestellnummer = rs.Fields("Bestell-Nr").Value()
            Debug.Print "Die Bestellnummer " & bestellnummer & " wurde nicht versandt."
        End If
        
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
    
End Sub

Sub vergleichDatenfeld_Verschachteln()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim bestellnummer As String
    Dim maxVersand As Date
    
    tblName = "Bestellungen"
    Set dbs = Application.CurrentDb
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    rs.MoveFirst
    Do While Not (rs.EOF)
        
        If IsNull(rs.Fields("Versanddatum").Value) Then
            
            maxVersand = DateAdd("m", 3, rs.Fields("Bestelldatum").Value)
           
            If (Date > maxVersand) Then
                bestellnummer = rs.Fields("Bestell-Nr").Value()
                Debug.Print "Die Bestellnummer " & bestellnummer & " wurde nicht versandt."
            End If
        End If
        
        rs.MoveNext
    Loop
    
    rs.Close
    
    Set rs = Nothing
    
    Set dbs = Nothing
    
End Sub

Sub vergleichObjektIsNothing()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim bestellnummer As String
    
    tblName = "Bestellungen"
    Set dbs = Application.CurrentDb
    
    If dbs Is Nothing Then
        Debug.Print "Kein Verweis vorhanden"
        Exit Sub
    End If
    
    Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
    
    If Not (rs Is Nothing) Then
          Do While Not (rs.EOF)
        
            If IsNull(rs.Fields("Versanddatum").Value) Then
                
                bestellnummer = rs.Fields("Bestell-Nr").Value()
                Debug.Print "Die Bestellnummer " & bestellnummer & " wurde nicht versandt."
            End If
            
            rs.MoveNext
        Loop
        
        rs.Close
    End If
    
    Set rs = Nothing
    
    Set dbs = Nothing
    
End Sub

Sub VerknuepfungUnd()
    Dim temperatur As Double
    
    temperatur = 39.8
    
    If (temperatur >= 36.9) And (temperatur <= 37.4) Then
        Debug.Print "Erhöhte Temperatur"
        
    ElseIf (temperatur >= 37.5) And (temperatur <= 39.4) Then
        Debug.Print "Fieber"
        
    ElseIf (temperatur >= 39.5) And (temperatur <= 40.5) Then
        Debug.Print "Hohes Fieber"
        
    ElseIf (temperatur >= 40.5) Then
        Debug.Print "Sehr hohes Fieber"
    
    Else
        Debug.Print "Normaltemperatur"
    End If
End Sub

Sub VerknuepfungOder()
    Dim monat As Integer
    
    monat = 3
    
    If (monat = 12) Or (monat = 1) Or (monat = 2) Then
        Debug.Print "Winter"
        
    ElseIf (monat = 3) Or (monat = 4) Or (monat = 5) Then
        Debug.Print "Frühjahr"
        
    ElseIf (monat = 6) Or (monat = 7) Or (monat = 8) Then
        Debug.Print "Sommer"
        
    ElseIf (monat = 9) Or (monat = 10) Or (monat = 11) Then
        Debug.Print "Herbst"
    
    Else
        Debug.Print "Keine korrekte Monatsangabe"
    End If
End Sub








Attribute VB_Name = "anweisung_Select"
Option Compare Database
Option Explicit

Sub beispiel_select_Ganzzahl()
    Dim monat As Byte
    
    monat = 6
    
    Select Case monat
        Case 3 To 5
            Debug.Print "Frühling"
        
        Case 6 To 8
            Debug.Print "Sommer"
            
        Case 9 To 11
            Debug.Print "Herbst"
            
        Case 12, 1, 2
            Debug.Print "Winter"
            
        Case Else
            Debug.Print "Kein Monat"
    End Select
End Sub

Sub beispiel_select_String()
    Dim land As String
    
    land = "Polen"
    
    Select Case land
        Case "Polen", "Dänemark", "Niederlande"
            Debug.Print "Europa"
        
        Case "Indien", "China"
            Debug.Print "Asien"
            
        Case Else
            Debug.Print "Unbekannter Kontinent"
    End Select
End Sub

Sub beispiel_select_Dezimal()
    Dim temperatur As Double
    
    ' temperatur = 39.8
    temperatur = 37.8
    
    Select Case temperatur
        ' Case Is >= 36.9 And temperatur <= 37.4 muss geschrieben werden als ..., andernfalls wird der case nicht korrekt ausgewertet
        Case 36.9 To 37.4
            Debug.Print temperatur & " = Erhöhte Temperatur"
        
        Case Is > 37.4 And temperatur <= 39.4
            Debug.Print temperatur & " = Fieber"
        
        Case Is > 39.4 And temperatur <= 40.5
            Debug.Print temperatur & " = Hohes Fieber"
        
        Case Is > 40.5
            Debug.Print temperatur & " = Sehr hohes Fieber"
    
        Case Else
            Debug.Print temperatur & " = Normaltemperatur"
    End Select
    
End Sub

Sub beispiel_Select_Date()
    Dim aktuellDatum As Date
    
    aktuellDatum = #12/24/2017#
    
    Select Case aktuellDatum
        Case #12/24/2017#
            Debug.Print "Heiligabend"
        Case #12/25/2017#
            Debug.Print "1. Weihnachtstag"
        Case #12/26/2017#
            Debug.Print "2. Weihnachtstag"
        Case #12/31/2017#
            Debug.Print "Silvester"
        Case #1/1/2018#
            Debug.Print "Neujahr"
        Case Else
            Debug.Print aktuellDatum
            
    End Select
    
End Sub


Sub beispiel_Select()
    Dim dbs As DAO.Database
    Dim rs As DAO.Recordset
    
    Dim tblName As String
    Dim bestellnummer As String
    
    Set dbs = Nothing
    Set rs = Nothing
    
    tblName = "Bestellungen"
    Set dbs = Application.CurrentDb
    
    If Not (dbs Is Nothing) Then
        Set rs = dbs.OpenRecordset(tblName, dbOpenTable)
        
        If Not (rs Is Nothing) Then
            rs.MoveFirst
            
            Do While Not (rs.EOF)
                
                Select Case rs.Fields("Status-Nr").Value
                    Case 0
                        Debug.Print "Die Bestellnummer " & rs.Fields("Bestell-Nr").Value & " wurde neu angelegt."
                    Case 1
                        Debug.Print "Die Bestellnummer " & rs.Fields("Bestell-Nr").Value & " wurde in Rechnung gestellt."
                    Case 2
                        Debug.Print "Die Bestellnummer " & rs.Fields("Bestell-Nr").Value & " wurde versandt."
                    Case 3
                        Debug.Print "Die Bestellnummer " & rs.Fields("Bestell-Nr").Value & " wurde geschlossen."
                    Case Else
                        Debug.Print "Die Bestellnummer " & rs.Fields("Bestell-Nr").Value & " ist ohne Statusangabe."
                    
                End Select
                
                rs.MoveNext
            Loop
    
            
            rs.Close
            
        End If
        
        Set rs = Nothing
        
    End If
    
    Set dbs = Nothing
    
End Sub


